


---

Rolf Müller

# Elektrische Systeme mit Maple analysieren - Komplexe Rechnung und Ortskurven

Zusatzmaterial zu



**Inhalt**

## 4.7 Komplexe Rechnung und Ortskurven

## 4.7.1 Rechnen mit komplexen Größen

## 4.7.2 Ortskurven

## 4.7.3 Komplexe Netzwerksberechnung mit dem Maschenstromverfahren

## 4 Modellierung elektrischer Netzwerke

### 4.7 Komplexe Wechselstromrechnung und Ortskurven

#### 4.7.1 Rechnen mit komplexen Größen

In den vorangegangenen Abschnitten des Kapitels 4 stand zwar die Berechnung von Ausgleichsvorgängen nach einer Änderung des Zustands von elektrischen Netzwerken im Vordergrund, die beschriebenen Rechnungen können aber auch Aussagen über die Netzwerkgrößen im stationären Zustand nach Ablauf des Einschwingvorgangs liefern, und zwar für beliebige Erregungen. Oft interessiert jedoch der Einschalt- oder Übergangsvorgang nicht, sondern nur der darauf folgende stationäre Zustand bei sinusförmigen Erregungen unter Beschränkung auf lineare Netzwerke. Für solche Fälle ist das Rechnen mit komplexen Größen - auch als symbolische Methode bezeichnet - ein sehr leistungsfähiges Verfahren. Bei dieser Art der Rechnung werden die sinusförmigen Wechselgrößen in den ohmschen Widerständen, Induktivitäten und Kapazitäten in entsprechende komplexe Größen abgebildet. Dadurch entstehen komplexe Operatoren, mit deren Hilfe komplexe Zeitfunktionen (rotierende Zeiger), komplexe Amplituden und komplexe Effektivwerte (ruhende Zeiger bei  $t = 0$ ) ineinander überführt werden können [Wei09]. Die für Gleichstromnetzwerke existierenden Gesetze (Abschnitt 4.1) gelten in analoger Form für das Rechnen mit den komplexen Größen.

**Tabelle 4.3** Netzwerkelemente und komplexe Impedanzen

Element	Zeitbereich	Bildbereich	$\underline{Z} = \underline{u}/\underline{i}$
Widerstand $R$	$u(t) = R \cdot i(t)$	$\underline{u} = R \cdot \underline{i}$	$\underline{Z}_R = R$
Induktivität $L$	$u(t) = L \cdot \frac{di(t)}{dt}$	$\underline{u} = j\omega \cdot L \cdot \underline{i}$	$\underline{Z}_L = j\omega \cdot L$
Kapazität $C$	$\frac{du(t)}{dt} = \frac{1}{C} \cdot i(t)$	$j\omega \cdot \underline{u} = \frac{1}{C} \cdot \underline{i}$	$\underline{Z}_C = \frac{1}{j\omega \cdot C} = -\frac{j}{\omega \cdot C}$


Die Angaben in obiger Tabelle haben Ähnlichkeit mit den Impedanz-Operatoren der Laplace-Transformation (Tabelle 4.2). An Stelle der Variablen  $j\omega$  erscheint bei den Laplace-Transformierten die Bildvariable  $s$ . Diese Ähnlichkeit darf aber nicht den unterschiedlichen Anwendungsbereich beider Verfahren vergessen lassen. Den zwei Verfahren gemeinsam ist, dass sie nur auf lineare Netzwerke anwendbar sind. Die mit der symbolischen Methode ermittelten Lösungen sind als Sonderfall in den mit der Laplace-Transformation ermittelten enthalten.

Als bekannt vorausgesetzt werden im Folgenden die Grundlagen der komplexen Berechnung von Wechselstromschaltungen (siehe z. B. [Froh11, Wei09]) und die Darlegungen des Kapitels 2.6 zum Arbeiten mit komplexen Zahlen und Zeigerdarstellungen in Maple. Anhand von Beispielen wird demonstriert, wie man dieses Computeralgebra-System beim Rechnen mit komplexen Größen in der Wechselstromtechnik sowie bei der Darstellung der Ergebnisse nutzen kann.

Komplexe Größen (Zeiger) werden in der Elektrotechnik heute meist durch einen Unterstrich gekennzeichnet. In den folgenden Maple-Programmen werden die entsprechenden Variablen für Spannungen und Ströme mit einem abschließenden Unterstrich versehen, wenn ohne diese Markierung eine Verwechslung mit dem Betrag der betreffenden Größe möglich wäre. Bei den komplexen Impedanzen wird auf diese Kennzeichnung verzichtet.

### Beispiel: Einfaches RC-Netzwerk

Die Spannung  $u_2(t)$  am Widerstand  $R_2$  und die komplexen Ströme in den Zweigen des Netzwerks des Bildes 4.17 sollen berechnet werden.



**Bild 4.1** Einfaches RC-Netzwerk, im rechten Bild mit komplexen Größen

Für die Eingangsspannung gilt

$$u(t) = \sqrt{2} \cdot U \cdot \cos(\omega \cdot t + \alpha)$$

Bei der Lösung der vorliegenden Aufgabe soll mit Effektivwerten gerechnet werden. Die komplexe Eingangsspannung ist somit

$$\underline{U} = U \cdot e^{j\alpha}$$

```
> restart: with(plots):
> interface(imaginaryunit = j):
```

Komplexe Spannung  $\underline{U}$

```
> G1:= U_ = U*exp(j*alpha);
```

Gleichung für die Berechnung der Impedanz  $Z_{par}$  der Parallelschaltung:

```
> G2:= Zpar = R2*1/(j*omega*C)/(R2+1/(j*omega*C));
```

$$G2 := Z_{par} = -\frac{j R2}{\omega C \left( R2 - \frac{j}{\omega C} \right)}$$

Die Beziehung für die Spannung  $U_2$  am Widerstand  $R_2$  lässt sich sehr einfach mit Hilfe der Spannungsteilerregel formulieren.

```
> G3:= U2_ = U_*Zpar/(R1+Zpar);
```

$$G3 := U2_- = \frac{U \cdot Z_{par}}{R1 + Z_{par}}$$

Gleichung G2 wird nun in die Gleichung G3 eingesetzt. Dabei sind die folgenden Annahmen für Maple wichtig, um eine möglichst einfache Lösung zu finden.

```
> assume(R1>0, R2>0, C>0, omega>0, alpha::real );
> G4:= subs(G2, G3);
```

$$G4 := U2\_ = - \frac{j U\_ R2}{\omega C \left( R2 - \frac{j}{\omega C} \right) \left( R1 - \frac{j R2}{\omega C \left( R2 - \frac{j}{\omega C} \right)} \right)}$$

```
> G4:= simplify(G4);
```

$$G4 := U2\_ = - \frac{j R2 U\_}{R1 R2 \omega C - j R1 - j R2}$$

Der Ausdruck für die komplexe Spannung  $\underline{U}$  wird in Gleichung G4 eingesetzt:

```
> G5:= subs(G1, G4);
```

$$G5 := U2\_ = - \frac{j R2 U e^{j\alpha}}{R1 R2 \omega C - j R1 - j R2}$$

Nun wird die rechte Seite der Gleichung G5 der Variablen U2\_ zugewiesen.

```
> assign(G5); U2_;
```

$$- \frac{j R2 U e^{j\alpha}}{R1 R2 \omega C - j R1 - j R2}$$

G5 wird für die folgende Rechnung nicht mehr benötigt. Es war daher akzeptabel, dass durch **assign** die linke Seite von G5 durch die rechte Seite überschrieben wurde.

Die nächsten zwei Anweisungen bestimmen den Betrag (Absolutwert) und den Winkel des Zeigers U2\_.

```
> U2:= abs(U2_);
```

$$U2 := \frac{R2 |U|}{\sqrt{R1^2 R2^2 \omega^2 C^2 + R1^2 + 2 R1 R2 + R2^2}}$$

```
> phi:= argument(U2_);
```

$$\phi := \text{argument} \left( - \frac{j U e^{j\alpha}}{R1 R2 \omega C - j R1 - j R2} \right)$$

Die komplexen Ströme in den Netzwerkzweigen liefern die folgenden Anweisungen:

```
> I2_:= U2_/R2; IC_:= U2_*j*omega*C; I_:= I2_+IC_;
```

$$I2\_ := - \frac{j U e^{j\alpha}}{R1 R2 \omega C - j R1 - j R2}$$

$$IC\_ := \frac{R2 U e^{j\alpha} \omega C}{R1 R2 \omega C - j R1 - j R2}$$

$$I_- := -\frac{j U e^{j\alpha}}{R_1 R_2 \omega C - j R_1 - j R_2} + \frac{R_2 U e^{j\alpha} \omega C}{R_1 R_2 \omega C - j R_1 - j R_2}$$

> `I_:= simplify(I_);`

$$I_- := \frac{U e^{j\alpha} (R_2 \omega C - j)}{R_1 R_2 \omega C - j R_1 - j R_2}$$

Damit im Folgenden die Auswertung der gefundenen Beziehungen für unterschiedliche Parameterwerte möglich ist, werden diese den Variablen (z. B.  $R$ ,  $C$  usw.) nicht fest zugewiesen, sondern in einer Parameterliste vereinbart. Mit dem Befehl **subs** werden anschließend die gesuchten Zahlenwerte berechnet.

> `param1:= R1=1, R2=2, C=0.001, omega=314, U=10, alpha=Pi/60;`

Komplexe Spannung am Widerstand  $R_2$ :

> `subs(param1, U2_);`

$$(6.386794493 - 1.336968981 j) e^{\frac{1}{60} j \pi}$$

Effektivwert  $U_2$ :

> `subs(param1, U2);`

$$0.6525230260 |10|$$

> `simplify(%);`

$$6.525230260$$

Winkel von  $\underline{U}_2$ : Berechnung zuerst im Bogenmaß und nach Definition der Funktion **grad** im Gradmaß.

> `subs(param1, phi);`

$$\text{argument}\left((31.93397247 - 6.684844903 j) e^{\frac{1}{60} j \pi}\right)$$

> `simplify(%);`

$$-0.1539937226$$

Definition der Funktion **grad** für die Umrechnung vom Bogen- ins Gradmaß:

> `grad:= arc -> simplify(180*arc/Pi);`

$$\text{grad} := \text{arc} \rightarrow \text{simplify}\left(\frac{180 \text{ arc}}{\pi}\right)$$

> `grad(subs(param1, phi));`

$$-8.823190375$$

Aus den berechneten Werten folgt

$$\underline{U}_2 = 6,525 \cdot e^{-j0,154}$$

bzw. nach Rücktransformation

$$u_2(t) = \sqrt{2} \cdot 6,525 \cdot \cos(\omega \cdot t - 8,8^\circ)$$

Für die komplexen Ströme in den Zweigen des Netzwerks ergeben sich folgende Werte:

```
> simplify(subs(param1, I2_));
3.224006582 - 0.5004388569 j
> simplify(subs(param1, IC_));
0.3142756023 + 2.024676133 j
> simplify(subs(param1, I_));
3.538282184 + 1.524237277 j
```


Das Zeigerbild für die berechneten komplexen Größen soll nun dargestellt werden. Um den dafür erforderlichen Schreibaufwand zu verringern bzw. die Übersichtlichkeit zu erhöhen, wird die Prozedur *zeiger* definiert.

```
> zeiger:= proc(x, ap, opt)
# Prozedur für einen Zeiger x mit dem Anfangspunkt ap
# opt...Optionen für Zeigerdarstellung (z.B. Farbe, Dicke)
local zeiger;
zeiger:= arrow([Re(ap), Im(ap)], [Re(x), Im(x)], op(opt));
end proc;
```

Dieser Prozedur sind die Optionen für die Zeigerdarstellung (*opt*) als Menge zu übergeben. Die darzustellende Größe wird über den formalen Parameter *x* übermittelt. Der Parameter *ap* legt den Anfangspunkt des Zeigers fest (siehe 2.6.). Die Prozedur gibt ein Plot-Konstrukt des Zeigers zurück. Dieses kann mit einem Befehl **display** dargestellt werden.

Bei den folgenden Aufrufen der Prozedur *zeiger* wird die Übergabe der darzustellenden Größe mit der Vorgabe der Parameterwerte durch den Befehl **subs** verbunden.

```
> Opt1:= {color=blue, width=0.1, head_length=0.5, head_width=0.3};
> zgI:= zeiger(subs(param1, I_), 0, Opt1);
> zgI2:= zeiger(subs(param1, I2_), 0, Opt1);
> zgIC:= zeiger(subs(param1, IC_), subs(param1, I2_), Opt1);
> display(zgI, zgIC, zgI2, scaling=constrained, axes=none);
```



Dieses Zeigerdiagramm der komplexen Ströme soll noch durch die Zeiger der Spannungen im Netzwerk und einige Bezeichnungen ergänzt werden.

Die Spannung am Widerstand  $R_1$  ist die Differenz der komplexen Spannungen  $\underline{U}$  und  $\underline{U}_2$ :

```
> U1_ := U - U2_;
```

$$U1_- := U_- + \frac{j R2 U e^{j\alpha}}{R1 R2 \omega C - j R1 - j R2}$$

```
> U1_:= subs(G1, U1_);
```


$$U1_ := U e^{j\alpha} + \frac{j R2 U e^{j\alpha}}{R1 R2 \omega C - j R1 - j R2}$$

```
> U1_:= simplify(U1_);
```

$$U1_ := \frac{U e^{j\alpha} R1 (R2 \omega C - j)}{R1 R2 \omega C - j R1 - j R2}$$

Erzeugung der Zeigerdarstellungen:

```
> Opt2:= {color=gray, width=0.1, head_length=0.5, head_width=0.3}:
> zgU:= zeiger(subs(param1,rhs(G1)), 0, Opt2):
> zgU2:= zeiger(subs(param1,U2_), 0, Opt2):
> zgU1:= zeiger(subs(param1,U1_), subs(param1,U2_), Opt2):
> bez:= textplot([[1.9,1.3,"I"], [1.8,-0.8,"I2"], [6.0,0.8,"U"],
[4.7,-1.2,"U2"]], font=[TIMES,14]):
> display(zgI, zgIC, zgI2, zgU, zgU2, zgU1, bez,
scaling=constrained, axes=none);
```



Die im Befehl **textplot** anzugebenden Positionen für die Darstellung der Texte kann man ermitteln, indem man die dargestellte Graphik markiert und den Mauszeiger an die gewünschte Stelle führt. Die Positionsangabe für diese Stelle wird dann in der Kontextleiste der Maple-Nutzeroberfläche (siehe 2.1.1) angezeigt.

#### 4.7.2 Ortskurven

##### Beispiel: Reihenschaltung R-L

```
> restart: with(plots):
> interface(imaginaryunit = j):
```

Komplexe Impedanz  $\underline{Z}$ :

```
> Z:= R + j*omega*L;
```

$$Z := R + j \omega L$$

```
> R:= 2: L:= 10^(-2):
```


```
> Z;
```

$$2 + \frac{1}{100} j \omega$$



Darstellung der Ortskurve der komplexen Impedanz  $Z$ :

```
> plot([Re(Z), Im(Z), omega=0..300], labels=[Re,Im],
       caption="Ortskurve von Z", color=blue);
```



Ermittlung der Ortskurve des komplexen Leitwerts  $Y$

```
> Y:= 1/Z;
```

$$Y := \frac{1}{2 + \frac{1}{100} j \omega}$$

$Y$  für  $\omega = 200$  Hz wird berechnet und zur Darstellung eines Zeigers (exakt: einer Linie) benutzt:

```
> Y200:= subs(omega=200, Y);
```


$$Y_{200} := \frac{1}{4} - \frac{1}{4} j$$

```
> zeig:= plot([[0,0],[Re(Y200),Im(Y200)]], color=black):
```

```
> pY:= plot([Re(Y), Im(Y), omega=0..infinity],
           labels=[Re,Im],caption="Ortskurve von Y",color=blue):
```

```
> bez:= textplot([0.26,-0.27, typeset(omega," = 200 Hz")]):
```

```
> display(pY, zeig, bez, scaling=constrained);
```



### 4.7.3 Komplexe Netzwerksberechnung mit dem Maschenstromverfahren

Wegen der schon genannten Analogie zwischen der komplexen Rechnung und der Netzbe-  
rechnung im Bildbereich der Laplace-Transformation kann hier das unter 4.5.2 beschriebene  
Berechnungsverfahren sinngemäß angewendet werden. Definiert man für jede Masche des  
vorgegebenen Netzwerks einen Maschenstrom, dann lässt sich für das Netzwerk gemäß Ab-  
schnitt 4.5.2 eine Matrizengleichung aufstellen.

$$\underline{\mathbf{Z}} \cdot \underline{\mathbf{I}} = \underline{\mathbf{U}} \quad \text{mit} \quad (4.16)$$

$\underline{\mathbf{Z}}$  ... Matrix der komplexen Maschenimpedanzen

$\underline{\mathbf{I}}$  ... Spaltenmatrix der komplexen Maschenströme

$\underline{\mathbf{U}}$  ... Spaltenmatrix der komplexen Quellenspannungen

Gleichung (4.16), nach der Spaltenmatrix der Maschenströme umgestellt, ergibt

$$\underline{\mathbf{I}} = \underline{\mathbf{Z}}^{-1} \cdot \underline{\mathbf{U}} \quad (4.17)$$

Über die Maschenströme können dann alle Zweigströme und Zweigspannungen berechnet  
werden. Ein einfaches Beispiel soll das Verfahren verdeutlichen.

#### Beispiel: Netzwerk mit drei Maschen

Für das in Bild 4.19 dargestellte Netzwerk mit der Spannung  $u(t) = \sqrt{2} \cdot U_1 \cdot \cos(\omega \cdot t + \alpha)$  sind die  
Maschenströme zu berechnen.




Bild 4.19 Netzwerk mit drei Maschen

Nach der Festlegung der Richtungen der Maschenströme (Bild 4.20) werden die komplexen  
Gesamtimpedanzen  $\underline{Z}_{11}$ ,  $\underline{Z}_{22}$  und  $\underline{Z}_{33}$  der drei Maschen sowie die komplexen Koppelimpedan-  
zen  $\underline{Z}_{i,k}$  abgelesen bzw. aufgeschrieben.




Bild 4.20 Netzwerk wie Bild 4.19 mit komplexen Maschenströmen

```

> restart:
> interface(imaginaryunit = j):

Komplexe Gesamtimpedanzen und Koppelimpedanzen:
> Z11:= R1+R2+j*omega*L1:
> Z22:= j*omega*L1+1/(j*omega*C1)+1/(j*omega*C2):
> Z33:= R2+j*omega*L2+1/(j*omega*C2):
> Z12:= -j*omega*L1:
> Z13:= -R2:
> Z23:= -1/(j*omega*C2):

```

Weil die Impedanzmatrix symmetrisch ist und damit  $Z_{i,k} = Z_{k,i}$  gilt, kann mit den dargestellten Impedanzen die Matrix  $\underline{Z}$  notiert werden:

```

> Z:= Matrix([[Z11,Z12,Z13],[Z12,Z22,Z23],[Z13,Z23,Z33]]);

```

$$Z := \begin{bmatrix} R1 + R2 + j \omega L1 & -j \omega L1 & -R2 \\ -j \omega L1 & j \omega L1 - \frac{j}{\omega C1} - \frac{j}{\omega C2} & \frac{j}{\omega C2} \\ -R2 & \frac{j}{\omega C2} & R2 + j \omega L2 - \frac{j}{\omega C2} \end{bmatrix}$$

Quellspannungsvektor und Maschenstromvektor:

```

> U:= Vector([U1_,0,0]);

```

$$U := \begin{bmatrix} U1_- \\ 0 \\ 0 \end{bmatrix}$$

```
> I:= Vector(' [I1_,I2_,I3_] ');
```

$$I := \begin{bmatrix} I1\_ \\ I2\_ \\ I3\_ \end{bmatrix}$$

Inverse der Impedanzmatrix  $\underline{Z}$  bestimmen:

```
> with(LinearAlgebra):
> Zinv:= MatrixInverse(Z):
```

Effektivwertzeiger der Spannung  $u(t)$ :

```
> U1_:= U1*exp(j*alpha):
```

Parameterwerte:

```
> param:= R1=3, R2=3, L1=1, L2=0.5, C1=0.01, C2=0.01, U1=10,
        alpha=45*Pi/180, omega=10:
```

Komplexen Vektor der Maschenströme berechnen:

```
> I:= Zinv.U:
```

Komplexe Maschenströme  $I_1$ ,  $I_2$  und  $I_3$ :

```
> I1_:=subs(param,I[1]); I2_:=subs(param,I[2]); I3_:=subs(param,I[3]);
```

$$I1_ := (1.804281346 - 0.4587155963j) e^{\frac{1}{4}j\pi}$$

$$I2_ := (1.529051988 + 0.4587155963j) e^{\frac{1}{4}j\pi}$$

$$I3_ := (3.333333333 - 0.j) e^{\frac{1}{4}j\pi}$$

Effektivwerte der Maschenströme:

```
> I1:= abs(I1_); I2:= abs(I2_); I3:= abs(I3_);
```

$$I1 := 1.861679665$$

$$I2 := 1.596377142$$

$$I3 := 3.333333333$$

Vor der Berechnung der Winkel der komplexen Ströme wird die Funktion *winkel* definiert, die den Winkel einer komplexen Größe  $x$  im Gradmaß liefert.

```
> winkel:= x -> evalf(180*argument(x)/Pi);
```

$$winkel := x \rightarrow evalf\left(\frac{180 \operatorname{argument}(x)}{\pi}\right)$$

```
> phi[1]:= winkel(I1_); phi[2]:= winkel(I2_); phi[3]:= winkel(I3_);
```

$$\phi_1 := 30.73548770$$

$$\phi_2 := 61.69924421$$

$$\phi_3 := 44.99999998$$

Damit können die gesuchten Maschenströme im Zeitbereich angegeben werden:

$$\begin{aligned} i_1(t) &\approx \sqrt{2} \cdot 1,86 \cdot \cos(10/s \cdot t + 30,7^\circ); & i_2(t) &\approx \sqrt{2} \cdot 1,6 \cdot \cos(10/s \cdot t + 61,7^\circ); \\ i_3(t) &\approx \sqrt{2} \cdot 3,33 \cdot \cos(10/s \cdot t + 45^\circ) \end{aligned}$$

### Literatur

Frohne, H. u. a.: Moeller Grundlagen der Elektrotechnik. 22. Aufl., Vieweg+Teubner Verlag 2011

Weißgerber, W.: Elektrotechnik für Ingenieure 2. Vieweg+Teubner Verlag 2009

Wunsch, G.: Systemanalyse. Bd. 1: VEB Verlag Technik Berlin

Ausgleichsvorgänge in elektro-mechanischen Systemen  
mit Maple analysieren

Grundwissen für Antriebstechnik und Mechatronik

Müller, R.

2011, XII, 284 S. 69 Abb. Mit zahlreichen Beispielen und  
Maple-Plots., Softcover

ISBN: 978-3-8348-1217-9